Three Level Models for Clustered Longitudinal Data (Time within Person within Twin Pair)

The data for this example come from the Octogenarian Twin Study of Aging, a longitudinal study (with 5 occasions spanning 8 years) of same-sex twin pairs initially age 79-100. We will be examining change over time in a measure of crystallized intelligence (information test), as well as prediction of that change from a measured of physical functioning (grip strength measured in pounds). These data are already stacked such that one row contains the data for one occasion for one person. The ID variables PairID and TwinID index which twin pair and which person, respectively, and Case is a unique identifier for each person. Time is unbalanced across persons, so the REPEATED statement will not be used (because we have to assume a VC R matrix anyway).

Model 1a: Empty Means, 2-Level Model for Information Test Outcome

$$\text{Level 1: } Info_{ti} = \beta_{0i} + e_{ti}$$
$$\text{Level 2: } \beta_{0i} = \gamma_{00} + U_{0i}$$

This model has 2 variance components: residual at level 1 and random intercept at level 2. It assumes that all people are independent (does not account for twin pair membership).

```
TITLE "SAS Model 1a: Empty Means, 2-Level Model for Information Test Outcome";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info =   / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID; RUN;

TITLE "SPSS Model 1a: Empty Means, 2-Level Model for Information Test Outcome".
MIXED info BY PairID TwinID
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  =
  /RANDOM = INTERCEPT | SUBJECT(PairID*TwinID) COVTYPE(UN).

 * STATA Model 1a: Empty Means, 2-Level Model for Information Test Outcome
xtmixed info ,  || Case:  , variance reml covariance(unstructured)
       estat ic, n(594)
       estimates store TwoLevel
```

SAS output:

```
           Dimensions
Covariance Parameters             2
Columns in X                      1
Columns in Z Per Subject          1
Subjects                        594 → number of persons so far
Max Obs Per Subject               5

                    Covariance Parameter Estimates
                                        Standard         Z
Cov Parm     Subject          Estimate       Error     Value      Pr > Z
UN(1,1)      PairID*TwinID      130.51      8.3822     15.57      <.0001
Residual                        26.6694      1.1203     23.81      <.0001

          Fit Statistics
-2 Res Log Likelihood         12147.4
AIC (smaller is better)       12151.4
AICC (smaller is better)      12151.4
BIC (smaller is better)       12160.2

  Null Model Likelihood Ratio Test
    DF    Chi-Square      Pr > ChiSq
     1       1411.30          <.0001

                  Solution for Fixed Effects
                       Standard
Effect       Estimate       Error      DF    t Value    Pr > |t|
Intercept     25.4630      0.4910     583      51.86      <.0001
```

Calculate the ICC for the proportion of between-person variation in Info:

$$ICC = \frac{130.52}{130.52 + 26.67} = .83$$

This LR test tells us that the random intercept variance is significantly greater than 0,

Model 1b: Empty Means, 3-Level Model for Information Test Outcome

Level 1: $Info_{tij} = \beta_{0ij} + e_{tij}$
Level 2: $\beta_{0ij} = \delta_{00j} + U_{0ij}$
Level 3: $\delta_{00j} = \gamma_{000} + V_{00j}$

This model now has 3 variance components: residual at level-1, random intercept at level 2, and random intercept at level 3. It now allows a correlation between people from the same twin pair.

```
TITLE "SAS Model 1b: Empty Means, 3-Level Model for Information Test Outcome";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info =   / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;           * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;    * Level 2; RUN;

TITLE "SPSS Model 1b: Empty Means, 3-Level Model for Information Test Outcome".
MIXED info BY PairID TwinID
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  =
  /RANDOM = INTERCEPT | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(PairID*TwinID) COVTYPE(UN).

 * STATA Model 1b: Empty Means, 3-Level Model for Information Test Outcome
xtmixed info ,  || PairID:  ,   covariance(unstructured) ///
       || Case: , variance reml covariance(unstructured)
       estat ic, n(337)
       estimates store ThreeLevel
       lrtest ThreeLevel TwoLevel
```

SAS output:

```
           Dimensions
Covariance Parameters             3
Columns in X                      1
Columns in Z Per Subject          3
Subjects                        337 → now number of twin pairs (families)
Max Obs Per Subject              10 → per twin pair (5 occasions * 2 persons)

                    Covariance Parameter Estimates
                                          Standard         Z
Cov Parm     Subject          Estimate       Error     Value      Pr > Z
UN(1,1)      PairID            83.7221      9.8155      8.53      <.0001 → level-3 between-pair
UN(1,1)      PairID*TwinID     47.3328      5.3992      8.77      <.0001 → level-2 within-pair
Residual                       26.7561      1.1270     23.74      <.0001 → level-1 within-person

          Fit Statistics
-2 Res Log Likelihood         12045.9
AIC (smaller is better)       12051.9
AICC (smaller is better)      12052.0
BIC (smaller is better)       12063.4
                  Solution for Fixed Effects
                       Standard
Effect       Estimate       Error      DF    t Value    Pr > |t|
Intercept     25.2102      0.5962     327      42.28      <.0001
```

Is the 3-level model a better fit than the 2-level model?
Yes, −2ΔLL(~1) = 101.5, p < .001

Proportion variance at each level:

Level 1 (time) = 26.75 / 157.83 = .17
Level 2 (person) = 47.34 / 157.83 = .30
Level 3 (pair) = 83.73 / 157.83 = .53

ICC_{L2} for time within person & pair =
(83.73 + 47.34) / (83.73 + 47.34 + 26.75) = .83

ICC_{L3} for person within pair = 83.72 / (83.72 + 47.33) = .64
This ICC = .64 is significantly greater than 0 via −2ΔLL for 3- vs. 2-level.

Now let's do the same thing for our two time-varying predictors: age and grip strength.

Age Model: Empty Means, 3-Level Model for Age Predictor

```
TITLE "SAS Age Model: Empty Means, 3-Level Model for Age Predictor";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL age =   / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;           * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;    * Level 2; RUN;

TITLE "SPSS Age Model: Empty Means, 3-Level Model for Age Predictor".
MIXED age BY PairID TwinID
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  =
  /RANDOM = INTERCEPT | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(PairID*TwinID) COVTYPE(UN).

 * STATA Age Model: Empty Means, 3-Level Model for Age Predictor
xtmixed age ,  || PairID:  ,   covariance(unstructured) ///
       || Case: , variance reml covariance(unstructured)
```

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	PairID	6.5528	0.6752	9.71	<.0001	level-3 between-pair = 47%
UN(1,1)	PairID*TwinID	0	.	.	.	level-2 within-pair = 0%
Residual		7.4662	0.2842	26.27	<.0001	level-1 within-person = 53%

Because there is no age variance at level 2, age will be a predictor at levels 1 and 3 only.

Fit Statistics

-2 Res Log Likelihood	8948.5
AIC (smaller is better)	8952.5
AICC (smaller is better)	8952.5
BIC (smaller is better)	8960.2

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	85.9648	0.1585	305	542.33	<.0001

Grip Strength Model: Empty Means, 3-Level Model for Grip Strength Predictor

```
TITLE "SAS Grip Model: Empty Means, 3-Level Model for Grip Strength Predictor";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL gripp =   / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;           * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;    * Level 2; RUN;

TITLE "SPSS Grip Model: Empty Means, 3-Level Model for Grip Strength Predictor".
MIXED gripp BY PairID TwinID
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  =
  /RANDOM = INTERCEPT | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(PairID*TwinID) COVTYPE(UN).

 * STATA Grip Model: Empty Means, 3-Level Model for Grip Strength Predictor
xtmixed gripp ,  || PairID:  ,   covariance(unstructured) ///
       || Case: , variance reml covariance(unstructured)
```

SAS output:

```
                  Covariance Parameter Estimates
                                       Standard         Z
Cov Parm     Subject          Estimate      Error     Value      Pr > Z
UN(1,1)      PairID             3.0858     0.4674      6.60      <.0001 level-3 between-pair = 36%
UN(1,1)      PairID*TwinID      2.5525     0.3437      7.43      <.0001 level-2 within-pair = 29%
Residual                        3.0496     0.1272     23.98      <.0001 level-1 within-person = 35%

          Fit Statistics
-2 Res Log Likelihood          7835.1
AIC (smaller is better)        7841.1
AICC (smaller is better)       7841.1
BIC (smaller is better)        7852.5

                  Solution for Fixed Effects
                       Standard
Effect       Estimate       Error      DF    t Value    Pr > |t|
Intercept      8.0660      0.1269     318      63.58      <.0001
```

Because grip strength has variance at all levels, grip strength will be a predictor at all levels. However, for simplicity we are ignoring for now its potential show person and pair differences in change over time.

We now need to create our predictor variables, including a mean of grip strength at the pair and person levels. We then code time as "time-in-study" and use baseline age as between-pair age. This gives us a convenient demarcation of age at baseline as the cross-sectional effect of age, and time-in-study as the longitudinal effect of age.

SAS Data Manipulation:

```
* Importing data into work library and creating person mean gripp for level-2;
DATA work.octodata; SET octo.octodata;
       PMgripp = MEAN(OF gripp1-gripp5);
       LABEL PMgripp= "PMgripp: Person Mean Gripp"; RUN;

* Getting twin pair means for grip strength to use at level-3;
PROC SORT DATA=work.octodata; BY PairID TwinID Wave; run;
PROC MEANS NOPRINT DATA=work.octodata; BY PairID; VAR PMgripp;
       OUTPUT OUT=PairMeans MEAN(PMgripp) = FMgripp; RUN;

* Merging PairMeans with datafile and centering predictors;
DATA work.octodata; MERGE work.octodata work.PairMeans; BY PairID;
       LABEL FMgripp= "FMgripp: Family Mean Gripp";

*** Age Variables ***;
       * Centering age at time 1 at 85 to use at level-3;
             BFage85 = agew1 - 85; LABEL BFage85= "BFage85: Age at Time1 (0=85)";
       * Within-person centering age at level-1 (like VARIABLE-BASED CENTERING);
             time = age - agew1;   LABEL time="time: Time Since Entry (0= Age Wave 1)";
*** Grip Strength Variables ***;
       * Centering family mean gripp at 9 to use at level-3;
             BFgripp9 = FMgripp - 9;
       * Centering person mean gripp at 9 to use at level-2;
             BPgripp9 = PMgripp - 9;          * CONSTANT-BASED CENTERING;
             WFgripp  = PMgripp - FMgripp;    * VARIABLE-BASED CENTERING;
       * Centering time-varying gripp to use at level-1;
             TVgripp9 = gripp - 9;            * CONSTANT-BASED CENTERING;
             WPgripp  = gripp - PMgripp;      * VARIABLE-BASED CENTERING;
       LABEL BFgripp9=  "BFgripp9:  Between-Family Mean Grip Strength in Pounds (0=9)"
             BPgripp9=  "BPgripp9: Between-Person Mean Grip Strength in Pounds (0=9)"
             WFgripp=   "WFgripp:  Within-Family Deviation from Mean Grip Strength in Pounds"
             TVgripp9=  "TVgripp9: Time-Varying Grip Strength in Pounds (0=9)"
             WPgripp=   "WPgripp:  Within-Person Deviation from Mean Grip Strength in Pounds";

* Selecting only cases with complete data;
       IF NMISS(agew1, age, FMgripp, PMgripp, gripp, info)>0 THEN DELETE; RUN;
```

SPSS Data Manipulation:

```
SORT CASES BY PairID TwinID Wave.
* Getting person gripp means to use as level-2 predictor.
COMPUTE PMgripp = MEAN(gripp1 TO gripp5).
EXECUTE.
* Getting pair gripp means to use as level-3 predictor.
AGGREGATE /OUTFILE=* MODE=ADDVARIABLES /PRESORTED /BREAK = PairID /FMgripp = MEAN(PMgripp).
VARIABLE LABELS FMgripp "FMgripp: Family Mean Gripp" PMgripp "PMgripp: Person Mean Gripp".
*** Age Variables ***.
      * Centering age at time 1 at 85 to use at level-3.
            COMPUTE BFage85 = agew1 - 85.
      * Within-person centering age at level-1 (like VARIABLE-BASED CENTERING).
            COMPUTE time = age - agew1.
            VARIABLE LABELS BFage85 "BFage85: Age at Time1 (0=85)"
                            time    "time: Time Since Entry (0= Age Wave 1)".
*** Grip Strength Variables ***.
      * Centering family mean gripp at 9 to use at level-3.
            COMPUTE BFgripp9 = FMgripp - 9.
      * Centering person mean gripp at 9 to use at level-2.
            COMPUTE BPgripp9 = PMgripp - 9.
            COMPUTE WFgripp  = PMgripp - FMgripp.
      * Centering time-varying gripp to use at level-1.
            COMPUTE TVgripp9 = gripp - 9.
            COMPUTE WPgripp  = gripp - PMgripp.
      VARIABLE LABELS
            BFgripp9 "BFgripp9: Between-Family Mean Grip Strength in Pounds (0=9)"
            BPgripp9 "BPgripp9: Between-Person Mean Grip Strength in Pounds (0=9)"
            WFgripp  "WFgripp:  Within-Family Deviation from Mean Grip Strength in Pounds"
            TVgripp9 "TVgripp9: Time-Varying Grip Strength in Pounds (0=9)"
            WPgripp  "WPgripp:  Within-Person Deviation from Mean Grip Strength in Pounds".
* Selecting only complete cases.
      SELECT IF (NMISS(agew1, age, FMgripp, PMgripp, gripp, info)=0).
      EXECUTE.
```

STATA Data Manipulation:

```
 * Creating person mean gripp for level-2
egen PMgripp = rmean(GRIPP1-GRIPP5)
label variable PMgripp "PMgripp: Person Mean Gripp"
 * Creating family mean gripp for level-3
egen FMgripp = mean(PMgripp), by(PairID)
label variable FMgripp "FMgripp: Family Mean Gripp"
 * Age variables
 * centering age at time 1 at 85 to use at level-3
gen BFage85 = agew1 - 85
label variable BFage85 "BFage85: Age at Time1 (0=85)"
 * within person centering age at level-1 (like VARIABLE-BASED CENTERING)
gen time = age - agew1
label variable time "time: Time since entry (0= Age Wave 1)"
 * Grip Strength Variables
 * centering family mean gripp at 9 use at level-3
gen BFgripp9 = FMgripp - 9
 * centering person mean gripp at 9 to use at level-2
gen BPgripp9 = PMgripp - 9          // CONSTANT-BASED CENTERING
gen WFgripp  = PMgripp - FMgripp    // VARIABLE-BASED CENTERING
 * centering time-varying gripp to use at level-1
gen TVgripp9 = gripp - 9            // CONSTANT-BASED CENTERING
gen WPgripp  = gripp - PMgripp      // VARIABLE-BASED CENTERING
label variable BFgripp9 "BFgripp9: Between-Family Mean Grip Strength in Pounds (0=9)"
label variable BPgripp9 "BPgripp9: Between-Person mean gripp strength in pounds (0=9)"
label variable WFgripp  "WFgripp:  Within-Family deviation from mean grip strength in Pounds"
label variable TVgripp9 "TVgripp9: Time-Varying Grip Strength in Pounds (0=9)"
label variable WPgripp  "WPgripp:  Within-Person Deviation from Mean Grip Strength in Pounds"

* Selecting only cases with complete data
egen nummiss = rowmiss(agew1 age FMgripp PMgripp gripp info)
drop if nummiss>0
```

Model 2a: Fixed Quadratic, Random Intercepts at Levels 2 and 3

Level 1: $\text{Info}_{tij} = \beta_{0ij} + \beta_{1ij}\left(\text{Age}_{tij} - \text{PairAge1}_j\right) + \beta_{2ij}\left(\text{Age}_{tij} - \text{PairAge1}_j\right)^2 + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(\text{PairAge1}_j - 85\right) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 2a: Fixed Quadratic, Random Intercept for Pair and Twin";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info = BFage85 time time*time / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID;           * Level 3;
       RANDOM INTERCEPT / TYPE=UN SUBJECT=PairID*TwinID;    * Level 2; RUN;

TITLE "SPSS Model 2a: Fixed Quadratic, Random Intercept for Pair and Twin".
MIXED info BY PairID TwinID WITH BFage85 time
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time
  /RANDOM = INTERCEPT | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT | SUBJECT(PairID*TwinID) COVTYPE(UN).

* STATA Model 2a: Fixed Quadratic, Random Intercepts at Levels 2 and 3
xtmixed info c.BFage85 c.time c.time#c.time , || PairID: , covariance(unstructured) ///
       || Case:  , variance reml covariance(unstructured)
       estat ic, n(337)
       estimates store FixQuad
```

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	PairID	79.5366	9.6947	8.20	<.0001
UN(1,1)	PairID*TwinID	52.4135	5.6798	9.23	<.0001
Residual		22.7722	0.9601	23.72	<.0001

Fit Statistics

-2 Res Log Likelihood	11878.0
AIC (smaller is better)	11884.0
AICC (smaller is better)	11884.1
BIC (smaller is better)	11895.5

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	25.1010	0.6835	378	36.73	<.0001
BFage85	-0.8074	0.1942	354	-4.16	<.0001
time	-0.2351	0.1457	1187	-1.61	0.1068
time*time	-0.05559	0.01872	1168	-2.97	0.0030

This model has 3 variance components: residual at level-1, random intercept at level-2, and random intercept at level-3. It now also has 3 new fixed effects: BFage85, time, and time^2.

We do not compare REML deviances because these models differ in fixed effects. Instead, we use their *p*-values. This is our new unconditional growth model baseline, as obtained from testing sequential models not shown here.

Model 2b: Fixed Quadratic, Random Linear Slope at Level 2

Level 1: $Info_{tij} = \beta_{0ij} + \beta_{1ij}\left(Age_{tij} - PairAge1_j\right) + \beta_{2ij}\left(Age_{tij} - PairAge1_j\right)^2 + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(PairAge1_j - 85\right) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 2b: Add Random Linear Slope for Twin";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info = BFage85 time time*time  / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT      / TYPE=UN SUBJECT=PairID;             * Level 3;
       RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;      * Level 2; RUN;

TITLE "SPSS Model 2b: Add Random Linear Slope for Twin".
MIXED info BY PairID TwinID WITH BFage85 time
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time
  /RANDOM = INTERCEPT      | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT time | SUBJECT(PairID*TwinID) COVTYPE(UN).

* STATA Model 2b: Add Random Linear Slope for Twin
xtmixed info c.BFage85 c.time c.time#c.time , || PairID: , covariance(unstructured) ///
       || Case: time , variance reml covariance(unstructured)
       estat ic, n(337)
       estimates store RandLin2
       lrtest RandLin2 FixQuad
```

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PairID	80.1040	9.4107	8.51	<.0001	→ level-3 intercept var
UN(1,1)	PairID*TwinID	44.3119	5.2577	8.43	<.0001	→ level-2 intercept var
UN(2,1)	PairID*TwinID	1.6220	0.7900	2.05	0.0401	→ level-2 int-linear cov
UN(2,2)	PairID*TwinID	1.1784	0.1806	6.53	<.0001	→ level-2 linear var
Residual		15.1230	0.8325	18.17	<.0001	→ level-1 residual var

Fit Statistics

-2 Res Log Likelihood	11746.0
AIC (smaller is better)	11756.0
AICC (smaller is better)	11756.0
BIC (smaller is better)	11775.1

This model has 2 new variance components at level 2: random linear slope and intercept-slope covariance.
Do we need the random linear slope for twin?
Yes, −2ΔLL(~2) = 132, p < .001

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	25.2772	0.6627	350	38.14	<.0001
BFage85	-0.7308	0.1909	347	-3.83	0.0002
time	-0.1455	0.1329	1168	-1.09	0.2741
time*time	-0.1021	0.01654	943	-6.17	<.0001

Model 2c: Fixed Quadratic, Random Linear Slope at Levels 2 and 3

Level 1: $Info_{tij} = \beta_{0ij} + \beta_{1ij}\left(Age_{tij} - PairAge1_j\right) + \beta_{2ij}\left(Age_{tij} - PairAge1_j\right)^2 + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(PairAge1_j - 85\right) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100} + V_{10j}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

```
TITLE "SAS Model 2c: Add Random Linear Slope for Pair";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info = BFage85 time time*time / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID;          * Level 3;
       RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;   * Level 2; RUN;

TITLE "SPSS Model 2c: Add Random Linear Slope for Pair".
MIXED info BY PairID TwinID WITH BFage85 time
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time
  /RANDOM = INTERCEPT time | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT time | SUBJECT(PairID*TwinID) COVTYPE(UN).

* STATA Model 2c: Add Random Linear Slope for Pair
xtmixed info c.BFage85 c.time c.time#c.time , || PairID: time, covariance(unstructured) ///
       || Case: time , variance reml covariance(unstructured)
       estat ic, n(337)
       estimates store RandLin23
       lrtest RandLin23 RandLin2
```

ICC of person within pair:
For Intercepts = 80.86 / (80.86 + 44.01) = .65
For Slopes = 0.06 / (0.06 + 1.12) = .05 (≈ 0)
Because the ICC for the slope at the pair level is not significantly different from 0, we will remove it.

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PairID	80.8615	9.5038	8.51	<.0001	→ level-3 intercept var
UN(2,1)	PairID	-0.7329	0.9258	-0.79	0.4286	→ level-3 int-linear cov
UN(2,2)	PairID	0.06408	0.1697	0.38	0.3529	→ level-3 linear var
UN(1,1)	PairID*TwinID	44.0073	5.2210	8.43	<.0001	→ level-2 intercept var
UN(2,1)	PairID*TwinID	1.9569	0.8826	2.22	0.0266	→ level-2 int-linear cov
UN(2,2)	PairID*TwinID	1.1164	0.2416	4.62	<.0001	→ level-2 linear var
Residual		15.1148	0.8311	18.19	<.0001	→ level-1 residual var

Fit Statistics

-2 Res Log Likelihood	11745.2
AIC (smaller is better)	11759.2
AICC (smaller is better)	11759.3
BIC (smaller is better)	11786.0

This model has 2 new variance components at level 3: random linear slope and intercept-slope covariance. **Do we need the random linear slope for pair, too?** *No, −2ΔLL(~2) = 0.8, p = .67*

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	25.2550	0.6639	348	38.04	<.0001
BFage85	-0.7439	0.1909	348	-3.90	0.0001
time	-0.1429	0.1333	1040	-1.07	0.2838
time*time	-0.1017	0.01654	944	-6.15	<.0001

TWO EQUIVALENT MODELS: VARIBLE-BASED CENTERING VS. CONSTANT-BASED CENTERING

Model 3a: Separate Effects of Grip Strength at Each Level via Variable-Based Centering

```
TITLE "SAS Model 3a: Grip Strength at each level via VARIABLE-BASED CENTERING";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
      CLASS PairID TwinID;
      MODEL info = BFage85 time time*time WPgripp WFgripp BFgripp9
                 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT      / TYPE=UN SUBJECT=PairID;            * Level 3;
      RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;     * Level 2;
      ESTIMATE "Level-2 Contextual Effect" WFgripp  1 WPgripp -1;
      ESTIMATE "Level-3 Contextual Effect" BFgripp9 1 WFgripp -1; RUN;

TITLE "SPSS Model 3a: Grip Strength at each level via VARIABLE-BASED CENTERING".
MIXED info BY PairID TwinID WITH BFage85 time WPgripp WFgripp BFgripp9
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time WPgripp WFgripp BFgripp9
  /RANDOM = INTERCEPT      | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT time | SUBJECT(PairID*TwinID) COVTYPE(UN)
  /TEST = "Level-2 Contextual Effect" WFgripp  1 WPgripp -1
  /TEST = "Level-3 Contextual Effect" BFgripp9 1 WFgripp -1.

* STATA Model 3a: Grip Strength at each level via VARIABLE-BASED CENTERING
xtmixed info c.BFage85 c.time c.time#c.time c.WPgripp c.WFgripp c.BFgripp9 , ///
    || PairID: , covariance(unstructured) || Case: time, variance reml covariance(unstructured)
      estat ic, n(337)
      lincom 1*c.WFgripp  - 1*c.WPgripp       // Level-2 Contextual Effect
      lincom 1*c.BFgripp9 - 1*c.WFgripp       // Level-3 Contextual Effect
```

Model 3b: Testing 3-Level Convergence of Grip Strength Effects via Constant-Based Centering

```
TITLE "SAS Model 3b: Grip Strength Convergence across levels via CONSTANT-BASED CENTERING";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
      CLASS PairID TwinID;
      MODEL info = BFage85 time time*time TVgripp9 BPgripp9 BFgripp9
                 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT      / TYPE=UN SUBJECT=PairID;            * Level 3;
      RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;     * Level 2;
      ESTIMATE "Level-2 Within-Family Effect" TVgripp9 1 BPgripp9 1;
      ESTIMATE "Level-3 Between-Pair  Effect" TVgripp9 1 BPgripp9 1 BFgripp9 1; RUN;

TITLE "SPSS Model 3b: Grip Strength Convergence across levels via CONSTANT-BASED CENTERING".
MIXED info BY PairID TwinID WITH BFage85 time TVgripp9 BPgripp9 BFgripp9
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time TVgripp9 BPgripp9 BFgripp9
  /RANDOM = INTERCEPT      | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT time | SUBJECT(PairID*TwinID) COVTYPE(UN)
  /TEST = "Level-2 Within-Family Effect" TVgripp9 1 BPgripp9 1
  /TEST = "Level-3 Between-Pair  Effect" TVgripp9 1 BPgripp9 1 BFgripp9 1.

* STATA Model 3b: Grip Strength Convergence across levels via CONSTANT-BASED CENTERING
xtmixed info c.BFage85 c.time c.time#c.time c.TVgripp9 c.BPgripp9 c.BFgripp9 , ///
    || PairID: , covariance(unstructured) || Case: time, variance reml covariance(unstructured)
      estat ic, n(337)
      lincom 1*c.TVgripp9 + 1*c.BPgripp9                  // Level-2 Within-Family Effect
      lincom 1*c.TVgripp9 + 1*c.BPgripp9 + 1*c.BFgripp9   // Level-3 Between-Pair Effect
```

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PairID	71.3908	8.5961	8.31	<.0001
UN(1,1)	PairID*TwinID	41.9006	5.0435	8.31	<.0001
UN(2,1)	PairID*TwinID	1.2241	0.7247	1.69	0.0912
UN(2,2)	PairID*TwinID	0.9945	0.1647	6.04	<.0001
Residual		15.3123	0.8413	18.20	<.0001

Because the models we will examine for grip strength are equivalent, the variance components and fit statistics are the same for both.

Fit Statistics

-2 Res Log Likelihood	11677.9
AIC (smaller is better)	11687.9
AICC (smaller is better)	11688.0
BIC (smaller is better)	11707.0

Model 3a: Separate Effects of Grip Strength at Each Level via Variable-Based Centering

Level 1: $\text{Info}_{tij} = \beta_{0ij} + \beta_{1ij}\left(\text{Age}_{tij} - \text{PairAge1}_j\right) + \beta_{2ij}\left(\text{Age}_{tij} - \text{PairAge1}_j\right)^2 + \beta_{3ij}\left(\text{Grip}_{tij} - \overline{\text{Grip}}_{ij}\right) + e_{tij}$

Within-person grip (WPgripp)

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + \delta_{01j}\left(\overline{\text{Grip}}_{ij} - \overline{\text{Grip}}_j\right) + U_{0ij}$

Within-family grip (WFgripp)

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Within-Person Grip: $\beta_{3ij} = \delta_{30j}$

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(\text{PairAge1}_j - 85\right) + \gamma_{002}\left(\overline{\text{Grip}}_j - 9\right) + V_{00j}$

Between-family grip (BFgripp9)

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

Within-Person Grip: $\delta_{30j} = \gamma_{300}$

Within-Family Grip: $\delta_{01j} = \gamma_{010}$

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	27.0432	0.7529	354	35.92	<.0001	
BFage85	-0.3463	0.1921	349	-1.80	0.0723	
time	0.08845	0.1386	1171	0.64	0.5235	
time*time	-0.1010	0.01653	954	-6.11	<.0001	
WPgripp	0.5031	0.09796	1184	5.14	<.0001	level-1, total within-person
WFgripp	0.9144	0.2251	281	4.06	<.0001	level-2, total within-family
BFgripp9	1.5114	0.2464	338	6.13	<.0001	level-3, total between-family

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Level-2 Contextual Effect	0.4112	0.2416	364	1.70	0.0895
Level-3 Contextual Effect	0.5971	0.3275	580	1.82	0.0688

Model 3b: Testing 3-Level Convergence of Grip Strength Effects via Constant-Based Centering

Level 1: $Info_{tij} = \beta_{0ij} + \beta_{1ij}\left(Age_{tij} - PairAge1_j\right) + \beta_{2ij}\left(Age_{tij} - PairAge1_j\right)^2 + \beta_{3ij}\left(Grip_{tij} - 9\right) + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + \delta_{01j}\left(\overline{Grip}_{ij} - 9\right) + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Within-Person Grip: $\beta_{3ij} = \delta_{30j}$

Within-person grip (TVgripp9)

Contextual between-person grip (BPgripp9)

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(PairAge1_j - 85\right) + \gamma_{002}\left(\overline{Grip}_j - 9\right) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

Within-Person Grip: $\delta_{30j} = \gamma_{300}$

Within-Family Grip: $\delta_{01j} = \gamma_{010}$

Contextual between-family grip (BFgripp9)

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	27.0432	0.7529	354	35.92	<.0001	
BFage85	-0.3463	0.1921	349	-1.80	0.0723	
time	0.08845	0.1386	1171	0.64	0.5235	
time*time	-0.1010	0.01653	954	-6.11	<.0001	
TVgripp9	0.5031	0.09796	1184	5.14	<.0001	level-1, total within-person
BPgripp9	0.4112	0.2416	364	1.70	0.0895	level-1 = level-2 effect?
BFgripp9	0.5971	0.3275	580	1.82	0.0688	level-2 = level-3 effect?

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Level-2 Within-Family Effect	0.9144	0.2251	281	4.06	<.0001
Level-3 Between-Pair Effect	1.5114	0.2464	338	6.13	<.0001

It appears that although there is a significant positive effect of grip strength at each level, those effects may not be significantly different in magnitude. Accordingly, let's simplify the model by removing the contextual effect at level 3, such that the level-2 and level-3 effects of grip strength are assumed to be the same.

Model 3c: Separate Effects of Grip Strength at Level 1 and Level-2&3 via Variable-Based Centering

```
TITLE "SAS Model 3c: Grip Strength at Level 1 and Level 2&3 via VARIABLE-BASED CENTERING";
PROC MIXED DATA=work.octodata NOCLPRINT NOITPRINT COVTEST NAMELEN=100 METHOD=REML;
       CLASS PairID TwinID;
       MODEL info = BFage85 time time*time WPgripp BPgripp9 / SOLUTION DDFM=Satterthwaite;
       RANDOM INTERCEPT      / TYPE=UN SUBJECT=PairID;                    * Level 3;
       RANDOM INTERCEPT time / TYPE=UN SUBJECT=PairID*TwinID;             * Level 2;
       ESTIMATE "Level-2&3 Contextual Effect" BFgripp9 1 WPgripp -1; RUN;

TITLE "SPSS Model 3c: Grip Strength at Level 1 and Level 2&3 via VARIABLE-BASED CENTERING".
MIXED info BY PairID TwinID WITH BFage85 time WPgripp BPgripp9
  /METHOD = REML
  /PRINT  = SOLUTION TESTCOV
  /FIXED  = BFage85 time time*time WPgripp BPgripp9
  /RANDOM = INTERCEPT      | SUBJECT(PairID) COVTYPE(UN)
  /RANDOM = INTERCEPT time | SUBJECT(PairID*TwinID) COVTYPE(UN)
  /TEST = "Level-2&3 Contextual Effect" BPgripp9 1 WPgripp -1.
```

```
* STATA Model 3c: Grip Strength at Level 1 and Level 2&3 via VARIABLE-BASED CENTERING
xtmixed info c.BFage85 c.time c.time#c.time c.WPgripp c.BPgripp9 , ///
    || PairID: , covariance(unstructured) || Case: time, variance reml covariance(unstructured)
      estat ic, n(337)
      lincom 1*c.BPgripp9 - 1*c.WPgripp       // Level-2&3 Contextual Effect
```

Level 1: $Info_{tij} = \beta_{0ij} + \beta_{1ij}\left(Age_{tij} - PairAge1_j\right) + \beta_{2ij}\left(Age_{tij} - PairAge1_j\right)^2 + \beta_{3ij}\left(Grip_{tij} - \overline{Grip}_{ij}\right) + e_{tij}$

Level 2:

Intercept: $\beta_{0ij} = \delta_{00j} + \delta_{01j}\left(\overline{Grip}_{ij} - 9\right) + U_{0ij}$

Linear Time: $\beta_{1ij} = \delta_{10j} + U_{1ij}$

Quadratic Time: $\beta_{2ij} = \delta_{20j}$

Within-Person Grip: $\beta_{3ij} = \delta_{30j}$

Within-person grip (WPgripp)

Between-person grip (BPgripp9)

Level 3:

Intercept: $\delta_{00j} = \gamma_{000} + \gamma_{001}\left(PairAge1_j - 85\right) + V_{00j}$

Linear Time: $\delta_{10j} = \gamma_{100}$

Quadratic Time: $\delta_{20j} = \gamma_{200}$

Within-Person Grip: $\delta_{30j} = \gamma_{300}$

Within-Family Grip: $\delta_{01j} = \gamma_{010}$

SAS output:

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	PairID	71.9633	8.6544	8.32	<.0001
UN(1,1)	PairID*TwinID	41.9783	5.0467	8.32	<.0001
UN(2,1)	PairID*TwinID	1.2345	0.7220	1.71	0.0873
UN(2,2)	PairID*TwinID	0.9953	0.1647	6.04	<.0001
Residual		15.3081	0.8409	18.21	<.0001

Fit Statistics

-2 Res Log Likelihood	11680.8
AIC (smaller is better)	11690.8
AICC (smaller is better)	11690.9
BIC (smaller is better)	11709.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	26.4767	0.6880	416	38.49	<.0001
BFage85	-0.4275	0.1874	366	-2.28	0.0231
time	0.09050	0.1386	1172	0.65	0.5139
time*time	-0.1011	0.01653	955	-6.12	<.0001
WPgripp	0.5071	0.09793	1185	5.18	<.0001
BPgripp9	1.1843	0.1696	556	6.98	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Level-2&3 Contextual Effect	0.6772	0.1926	849	3.52	0.0005

One could then test interactions, making sure to differentiate effects across all three levels as needed…

Sample Results Section (note this combines across models somewhat)

The extent of individual change in crystallized intelligence (as measured by the information test) and the relationship between intelligence, age, and grip strength was examined in a sample of 337 same-sex twins measured every two years for up to five occasions. Multilevel models were estimated using restricted maximum likelihood. The significance of fixed effects was evaluated with individual Wald tests (i.e., of estimate / SE), whereas random effects were evaluated via likelihood ratio tests (i.e., $-2\Delta LL$ with degrees of freedom equal to the number of new random effects variances and covariances).

A two-level empty means, random intercept model of time nested within person was initially specified and indicated that 83% of the information test outcome variance was between persons. The addition of a random intercept for twin pair resulted in a significant improvement in model fit, $-2\Delta LL(1) = 101.5$, $p < .001$, and revealed that 64% of that between-person variance was due to twin pair (i.e., shared variance between twins from the same pair). Thus, a three-level model was necessary, given that 17% of the variance was at level 1 (within persons over time), 30% was at level 2 (within pairs), and 53% was at level 3 (between pairs). A three-level empty means, random intercept model to decompose the variance in time-varying age revealed that 47% was between pairs (given that the twins initially varied in age from 80 to 100), whereas the remaining 53% was within persons over time—there was no level-2 age variance. Thus, the level-3 cross-sectional and level-1 longitudinal effects of age were modeled separately using baseline age (centered at 85) and time in study, respectively. Preliminary analyses revealed that a linear effect of age at baseline and a quadratic effect of time in study resulted in the best-fitting model to describe mean change. Although a random linear time slope for twin significantly improved model fit, $-2\Delta LL(2) = 132.0$, $p < .001$, the subsequent addition of a random linear time slope for twin pair did not significantly improve model fit, $-2\Delta LL(2) = 0.8$, $p = .67$, indicating that the 5% of the random linear time slope variance that was due to twin pair was not distinguishable from 0. As a result, the random linear time slope was retained at the twin level only (i.e., level 2 but not level 3).

The prediction of the information test outcome from time-varying grip strength was then examined. A three-level empty means, random intercept model to decompose the variance in grip strength revealed that 36% was between pairs, 29% was within pairs, and 35% was within persons over time. Predictors for grip strength were included via variable-based centering, in which the within-person effect was represented by the deviation of each occasion's grip strength around each person's mean, the within-pair effect was represented by the deviation of each twin's mean grip strength around each pair's mean, and the between-pair effect was represented by the family mean grip strength (centered at 9 pounds). There was a significant main effect of grip strength at each level. Within persons, for every additional pound of grip strength more than one's own mean, information test at that occasion was expected to be higher by 0.50. Within pairs, for every additional pound of person mean grip strength more than one's family mean, information test for that twin was expected to be higher by 0.91. Between pairs, for every additional pound of family mean grip strength more than other families, information test for the twin pair was expected to be higher by 1.51.

Contextual effects for the differences in effect size across levels were requested using separate statements (i.e., as would be provided directly using constant-based centering but including the person and pair means). The pair-level contextual effect was not significant, indicating that the within-pair and between-pair effects were equivalent. Consequently, the model was re-specified to include within-person grip strength, as described previously, along with between-person grip strength to represent the combination of the twin and pair levels, calculated as each person's mean grip strength centered at 9. The between-person effect of grip strength was significant, such that for every additional pound of mean grip strength more than other people, information test for that twin was expected to be higher by 1.18. This effect was significantly larger than the within-person effect of grip strength of 0.51 (i.e., a significant person contextual effect), and thus both the within-person and between-person effects of grip strength were retained.